
Introduction

There has been so much talk about global warming.

Is it real?

Is it human-made?

Global warming starts at home.

So, let’s address the question of local warming.

Has it been getting warmer in NJ?



The Data

The National Oceanic and Atmospheric Administration (NOAA) collects and archives weather
data from thousands of collection sites around the globe. The data format and instructions
on how to download the data can be found on this NOAA website:

ftp://ftp.ncdc.noaa.gov/pub/data/gsod/readme.txt

The list of the roughly 9000 weather stations is posted here:

ftp://ftp.ncdc.noaa.gov/pub/data/gsod/ish-history.txt

Perusing this list, I discovered that McGuire Air Force Base, located not far from Princeton
NJ, is one of the archived weather stations. Since, the data is archived in one year batches,
I wrote a unix shell script to grab the 55 annual data files for McGuire and then assemble
the relevant pieces of data into a single file. Here is the shell script...

http://www.princeton.edu/∼rvdb/ampl/nlmodels/LocalWarming/McGuireAFB/data/getData.sh

The resulting pair of data files that I used as input to my local climate model are posted at...

http://www.princeton.edu/∼rvdb/ampl/nlmodels/LocalWarming/McGuireAFB/data/McGuireAFB.dat

and

http://www.princeton.edu/∼rvdb/ampl/nlmodels/LocalWarming/McGuireAFB/data/Dates.dat

ftp://ftp.ncdc.noaa.gov/pub/data/gsod/readme.txt
ftp://ftp.ncdc.noaa.gov/pub/data/gsod/ish-history.txt
http://www.princeton.edu/~rvdb/ampl/nlmodels/LocalWarming/McGuireAFB/data/getData.sh
http://www.princeton.edu/~rvdb/ampl/nlmodels/LocalWarming/McGuireAFB/data/McGuireAFB.dat
http://www.princeton.edu/~rvdb/ampl/nlmodels/LocalWarming/McGuireAFB/data/Dates.dat


McGuire AFB







The Model

Let Td denote the average temperature in degrees Fahrenheit on day d ∈ D where D is the
set of days from January 1, 1955, to August 13, 2010 (that’s 20,309 days).

We wish to model the average temperature as a constant a0 plus a linear trend a1d plus a
sinusoidal function with a one-year period representing seasonal changes,

a2 cos(2πd/365.25) + a3 sin(2πd/365.25),

plus a sinusoidal function with a period approximately equal to 10.7 years to represent the
solar cycle,

a4 cos(2a6πd/(10.7× 365.25)) + a5 sin(2a6πd/(10.7× 365.25)).

The parameters a0, a1, . . . , a6 are unknown regression coefficients. We wish to find the values
of these parameters that minimize the sum of the absolute deviations:

min
a0,...,a6

∑
d∈D

| a0 + a1d

+a2 cos(2πd/365.25) + a3 sin(2πd/365.25)

+a4 cos(a62πd/(10.7× 365.25)) + a5 sin(a62πd/(10.7× 365.25))

−Td | .



Linearizing the Solar Cycle

If the unknown parameter a6 is fixed at 1, forcing the solar-cycle to have a period of exactly
10.7 years, then the problem can be reduced to a linear programming problem.

If, on the other hand, we allow a6 to vary, then the problem is nonlinear and even nonconvex
and therefore much harder in principle. Nonetheless, if we initialize a6 to one, then the
problem might, and in fact does, prove to be tractable.

Note: This least-absolute-deviations (LAD) model automatically ignores “outliers” such as
the record heat wave of 2010.



AMPL Model
set DATES ordered;
param avg {DATES};
param day {DATES};
param pi := 4*atan(1);

var a {j in 0..6};
var dev {DATES} >= 0, := 1;

minimize sumdev: sum {d in DATES} dev[d];
subject to def_pos_dev {d in DATES}:

a[0] + a[1]*day[d] + a[2]*cos( 2*pi*day[d]/365.25)
+ a[3]*sin( 2*pi*day[d]/365.25)
+ a[4]*cos( a[6]*2*pi*day[d]/(10.7*365.25))
+ a[5]*sin( a[6]*2*pi*day[d]/(10.7*365.25))

- avg[d]
<= dev[d];

subject to def_neg_dev {d in DATES}:
-dev[d] <=
a[0] + a[1]*day[d] + a[2]*cos( 2*pi*day[d]/365.25)

+ a[3]*sin( 2*pi*day[d]/365.25)
+ a[4]*cos( a[6]*2*pi*day[d]/(10.7*365.25))
+ a[5]*sin( a[6]*2*pi*day[d]/(10.7*365.25))

- avg[d];



AMPL Data and Variable Initialization

data;

set DATES := include "data/Dates.dat";
param: avg := include "data/McGuireAFB.dat";
let {d in DATES} day[d] := ord(d,DATES);

let a[0] := 60;
let a[1] := 0;
let a[2] := 20;
let a[3] := 20;
let a[4] := 0.01;
let a[5] := 0.01;
let a[6] := 1;

The nice thing about ampl and loqo is that anyone can use these programs via the NEOS
server at Argonne National Labs...

http://www-neos.mcs.anl.gov/

http://www-neos.mcs.anl.gov/


The Results

The linear version of the problem solves in a small number of iterations and only takes
a minute or so on my MacBook Pro laptop computer. The nonlinear version takes more
iterations and more time but eventually converges to a solution that is almost identical to
the solution of the linear version. The optimal values of the parameters are

a0 = 52.6 ◦F

a1 = 9.95× 10−5 ◦F/day

a2 = −20.4 ◦F
a3 = −8.31 ◦F
a4 = −0.197 ◦F
a5 = 0.211 ◦F

a6 = 0.992

From a0, we see that the nominal temperature at McGuire AFB was 52.56 ◦F (on January
1, 1955).

We also see, from a1, that there is a positive trend of 0.000099 ◦F/day. That translates to
3.63 ◦F per century—in amazing agreement with results from global climate change models.

A 95% confidence interval for a1 is [3.6132 ◦F, 3.7133 ◦F]/100 yrs.



Magnitude of the Sinusoidal Fluctuations

From a2 and a3, we can compute the amplitude of annual seasonal changes in temperatures...√
a22 + a23 = 22.02 ◦F.

In other words, on the hottest summer day we should expect the temperature to be 22.02
degrees warmer than the nominal value of 52.56 degrees; that is, 77.58 degrees. Of course,
this is a daily average—daytime highs will be higher and nighttime lows should be about the
same amount lower.

Similarly, from a4 and a5, we can compute the amplitude of the temperature changes brought
about by the solar-cycle... √

a24 + a25 = 0.2887 ◦F.

The effect of the solar cycle is real but relatively small.

The fact that a6 came out slightly less than one indicates that the solar cycle is slightly longer
than the nominal 10.7 years. It’s closer to 10.7/a6 = 10.78 years.



Plot Showing Actual Data and Regression Curve

Blue: Average daily temperatures at McGuire AFB from 1955 to 2010.
Red: Output from least absolute deviation regression model.

Seasonal fluctuations completely dominate other effects.



Subtracting Out Seasonal Effects

As before but with sinusoidal seasonal variation removed and sinusoidal solar-cycle variation
removed as well.

Even this plot is noisy simply because there are many days in a year and some days are
unseasonably warm while others are unseasonably cool.



Smoothed Seasonally Subtracted Plot

To smooth out high frequency fluctuations, we use 101 day rolling averages of the data.

In this plot, the long term trend in temperature is clearly seen. In NJ we have local warming.



Means, Medians, and Optimization

Let b1, b2, . . . , bn denote a set of measurements.

Solving
argminx

∑
i

(x− bi)2

computes the mean.

Solving
argminx

∑
i

|x− bi|

computes the median.

Medians correspond to nonparametric statistics. Nonparametric confidence intervals are
given by percentiles. The p-th percentile is computed by solving the following optimization
problem:

argminx

∑
i

(
|x− bi| + (1− 2p)(x− bi)

)
.



Confidence Intervals For Medians

Assume that B1, B2, B3, . . . , Bn are independent identically distributed with median m.

Let
B(1) < B(2) < B(3) < · · · < B(n)

denote the order statistics, i.e., the original variables rearranged into increasing order.

Note: B(k) is the (k/n)-th sample percentile.

Then,

P(B(k) ≤ m ≤ B(k+1)) = P(Bj ≤ m for k indices and

Bj ≥ m for the remaining n− k indices)

=

(
n
k

)(
1

2

)n

.

Hence,

P(B(k) ≤ m ≤ B(n−k+1)) =
n−k∑
j=k

(
n
j

)(
1

2

)n

.

For any given n, it is easy to choose k so that
n−k∑
j=k

(
n
j

)(
1

2

)n

≈ 0.95.



Relaxing the Length of a Year

The number 365.25 days per year is not exactly correct.

One extra variable, a[7], can be added to the model. It is analogous to a[6].

The new variable makes the problem even more nonconvex but miraculously it becomes easier
to solve.

Here’s the output:

a0 = 52.5 ◦F

a1 = 1.02× 10−4 ◦F/day

a2 = −20.1 ◦F
a3 = −9.02 ◦F
a4 = −0.194 ◦F
a5 = 0.181 ◦F

a6 = 0.998

a7 = 1.00021

The correct value of the correction factor a7 is 1.0000068 (one day every 400 years). The
value of a7 from the regression model makes a correction in the right direction but it is too
big. Clearly this effect was too small to pull out of the data.



Least Squares Solution (Mean instead of Median)

Suppose we change the objective to a sum of squares of deviations:

minimize sumdev: sum {d in DATES} dev[d]^2;

The resulting model is a least squares model.
The objective function is now convex and quadratic and the problem is still easy to solve.
The solution, however, is sensitive to outliers.
Here’s the output:

a0 = 52.6 ◦F

a1 = 1.2× 10−4 ◦F/day

a2 = −20.3 ◦F
a3 = −7.97 ◦F
a4 = 0.275 ◦F

a5 = 0.454 ◦F

a6 = 0.730

In this case, the rate of local warming is 4.37 ◦F per century.
However, the model produces the wrong answer for the period of the solar cycle.



Final Remarks

Close inspection of the output shows that the January 22 is the coldest day in the winter,
July 24 is nominally the hottest day of summer, and February 12, 2007, was the day of the
last minimum in the 10.78 year solar cycle.

The ampl model and the shell scripts are available on my webpage.

Everyone is encouraged to grab data for any location they like.

Send me the results and I’ll compute a global average.


